

Irrationality and Transcendence Criteria for Infinite Series in Isabelle/HOL

Angeliki Koutsoukou-Argyraki (ak2110@cam.ac.uk)
Wenda Li (wl302@cam.ac.uk)
Lawrence C. Paulson FRS (lp15@cam.ac.uk)

Computer Laboratory, University of Cambridge
15 JJ Thomson Avenue, Cambridge CB3 0FD, UK

April 27, 2021

Abstract

We give an overview of our formalizations in the proof assistant Isabelle/HOL of certain irrationality and transcendence criteria for infinite series from three different research papers: by Erdős and Straus (1974), Hančl (2002) and Hančl and Rucki (2005). Our formalizations in Isabelle/HOL can be found on the Archive of Formal Proofs. Here we describe selected aspects of the formalization and discuss what this reveals about the use and potential of Isabelle/HOL in formalizing modern mathematical research, particularly in these parts of number theory and analysis.

Keywords: transcendence, irrationality, series, interactive theorem proving, Isabelle/HOL, proof assistants.

AMS 2020 subject classes: 40A05, 11J68, 11J81, 03B35, 68V20, 68V35.

1 Introduction

The libraries of formalized mathematics in various proof assistants (or interactive theorem provers) have been growing for years. However, the field

is still far from the goal propounded in 1994 as the QED Manifesto [5] “to build a computer system that effectively represents all important mathematical knowledge and techniques”. We are light years from the vision of having an interactive tool that could “converse” with human mathematicians to aid in the discovery of new results, as colorfully described by Timothy Gowers [17]. Our goal within the ALEXANDRIA Project at Cambridge [37] is to contribute to making proof assistants (and more specifically Isabelle/HOL) more useful to research mathematicians. To gather information about what specifically needs to be done, we have been formalizing material to contribute to the Isabelle/HOL Libraries¹ and the Archive of Formal Proofs (AFP).² As research mathematics is our main target, and given that a considerable amount of basic mathematics has already been formalized in Isabelle/HOL over the years, here we wish to explore the following **question**:

How receptive to formalization (particularly in Isabelle/HOL) is mainstream published mathematical research?

In this paper we are not working towards the ambitious goal of formalizing the proof of a landmark result. Nor are we concerned with standard material from undergraduate or graduate textbooks. Those proofs have been thoroughly studied and checked, and repeatedly reworked and polished by many people over the years. Instead, we are formalizing mainstream mathematical research from journal papers in their raw form, not reworked or clarified as textbook proofs normally are. To explore the question above (within a specific case study), we ask questions such as these:

- To what extent does the current Isabelle/HOL library cover the necessary preliminaries?
- How helpful are the automatic proof tools?
- How much longer than the original proofs are the formalized proofs (i.e. the *de Bruijn factor* [40])?
- Are we going to discover any issues or gaps in the proofs?

Clearly, the answers to these questions depend on the chosen proofs. Here we attempt some case studies in number theory, on transcendence and irrationality criteria for infinite series. In particular, we choose several results

¹isabelle.in.tum.de/dist/library/HOL

²www.isa-afp.org

from three research papers by Erdős and Straus (1974) [13], Hančl (2002) [23], and Hančl and Rucki (2005) [24]. This line of research makes heavy use of standard background material from analysis, e.g. convergence tests for series and (for Erdős and Straus [13]) certain results about prime numbers. The material from Hančl and Rucki [24] applies Roth’s theorem on rational approximations to algebraic numbers.³ Inevitably, we had to pick topics in which we a priori knew that at least some of the preliminaries were already available in Isabelle/HOL. The material we chose is mathematically interesting in its own right, and moreover—unlike most results formalized nowadays—it comes from fairly recent research papers instead of textbooks.

Paul Erdős was keenly interested in establishing criteria for the irrationality and transcendence of infinite series [11, 12, 14]. Several techniques for showing the transcendence of series are given by Nishioka [33], who presents, among other, methods originated by Mahler [31]. Later, Sándor [39], Nyblom [34, 35], and Hančl [21, 22, 20] obtained many more results in this area.

In this paper, we merely present in standard mathematical language the statements (without the informal proofs) of the theorems that we formalized and afterwords we discuss our formalization in Isabelle/HOL focussing on its most interesting aspects. For the informal proofs the reader may refer to the respective papers [13, 23, 24]. Our full Isabelle/HOL formalizations can be found online, on the AFP [30, 28, 29]. As we had to fill in many intermediate arguments that were implicit in the original proofs, providing more details and clarifications, our formalizations may serve not only as case studies to explore the question of how receptive to formalization in Isabelle/HOL mathematical research is and as proof verification, but also as helpful supplementary material for a detailed study of the original proofs.

The plan of this paper is as follows: in the next section we give a brief introduction to Isabelle/HOL. In Sections 3, 4 and 5 we present the material formalized and discuss the main challenges and lessons learned through our formalizations for each paper respectively. In Section 6 we briefly discuss the de Bruijn factors of the formalizations and finally, in Section 7 we present our conclusion.

³The proof of Roth’s theorem itself has not been formalized; see Section 5.

2 Isabelle/HOL

Isabelle⁴ is an interactive theorem prover first developed in the 1980s by Lawrence Paulson and Tobias Nipkow [32, 36]. Today, Isabelle provides the *Isar* language for writing proofs. Isar proofs are hierarchically structured, and they include enough redundancy—consisting of explicitly stated assumptions and goals—to be understandable to humans as well as machines. Isabelle supports multiple different logical formalisms, with Isabelle/HOL specifically based on higher order logic. Unlike proof assistants based on constructive type theories with dependent types, it implements simple type theory. Isabelle/HOL admits classical (non-constructive) proofs, i.e. it accepts the law of the excluded middle, $P \vee \neg P$, as well as the axiom of choice.

Simple type theory and classical logic allow for powerful automation, which is a significant advantage. Isabelle/HOL’s main automation tool is *Sledgehammer* [2], which searches for proofs by calling external automated theorem provers, and then converts the proofs discovered to Isar. Two counterexample finding tools, *nitpick* and *Quickcheck*, are also provided. Moreover, the command `try0` calls a number of built-in proof methods (such as simplification) in search of a proof. The user interface is Isabelle/jEdit, which provides real-time proof checking, as well as rich semantic information for the formal text and direct links to the user manuals and tutorials.⁵ We have written an introduction to Isabelle aimed at mathematicians [27].

The Isabelle/HOL libraries and the AFP contain a vast amount of formalized material, covering many areas of pure mathematics as well as in computer science, logic and even philosophy.

3 On the irrationality of certain series

3.1 Results formalized

The following results are by Erdős and Straus [13].

Theorem 3.1 (Erdős and Straus [13, Theorem 2.1]) *Let $\{b_n\}_{n=1}^{\infty}$ be a sequence of integers and $\{\alpha_n\}_{n=1}^{\infty}$ a sequence of positive integers with $\alpha_n > 1$*

⁴<https://isabelle.in.tum.de/>

⁵These can also be found under “Documentation” on the Isabelle webpage.

for all large n and

$$\lim_{n=1, n \rightarrow \infty} \frac{|b_n|}{\alpha_{n-1}\alpha_n} = 0.$$

Then the sum

$$\sum_{n=1}^{\infty} \frac{b_n}{\prod_{i=1}^n \alpha_i}$$

is rational if and only if there exists a positive integer B and a sequence of integers $\{c_n\}_{n=1}^{\infty}$ so that for all large n ,

$$Bb_n = c_n\alpha_n - c_{n+1}, \quad |c_{n+1}| < \alpha_n/2.$$

Corollary 3.1 (Erdős and Straus [13, Corollary 2.10]) *Let $\{\alpha_n\}_{n=1}^{\infty}$ and $\{b_n\}_{n=1}^{\infty}$ satisfy the hypotheses of the theorem above and in addition that for all large n we have $b_n > 0$, $\alpha_{n+1} \geq \alpha_n$, $\lim_{n \rightarrow \infty} (b_{n+1} - b_n)/\alpha_n \leq 0$ and $\liminf_{n \rightarrow \infty} \alpha_n/b_n = 0$. Then the sum*

$$\sum_{n=1}^{\infty} \frac{b_n}{\prod_{i=1}^n \alpha_i}$$

is irrational.

Theorem 3.2 (Erdős and Straus [13, Theorem 3.1]) *Let p_n be the n th prime number and let $\{\alpha_n\}_{n=1}^{\infty}$ be a monotonic sequence of positive integers satisfying $\lim_{n \rightarrow \infty} p_n/\alpha_n^2 = 0$ and $\liminf_{n \rightarrow \infty} \alpha_n/p_n = 0$. Then the sum*

$$\sum_{n=1}^{\infty} \frac{p_n}{\prod_{i=1}^n \alpha_i}$$

is irrational.

3.2 On the formalization

Our full formalization is online [30] and here we discuss certain key points. The formalization process turned out to be time-consuming: we had to fill in a number of intermediate reasoning steps that had to be made explicit in the formal proofs. For example, in the proof of Theorem 3.2 ([13, Theorem 3.1]), for the sequence of integers $\{c_n\}_{n=1}^{\infty}$ (for which $c_n > 0$ for large n), Erdős and Straus claim that since it is unbounded, for large n there must exist an index

$m \geq n$ so that $c_m \leq c_n < c_{m+1}$. In order to prove this claim, we had first to show that for large n the sequence $\{c_n\}_{n=1}^\infty$ is neither monotone increasing nor monotone decreasing, which required some work.

We moreover had to do a certain amount of restructuring for the formalization of the proof of Theorem 3.1 ([13, Theorem 2.1]), as the original proof is based on a sketch of a construction for obtaining the terms of the desired sequence $\{c_n\}_{n=1}^\infty$, starting with the construction of the first two terms for large enough n (“Proceeding in this manner we get the desired sequence”). In particular, in the sufficient direction, we need to obtain a positive integer B and a sequence of integers $\{c_n\}_{n=1}^\infty$ that satisfy $Bb_n = c_n\alpha_n - c_{n+1}$ and $|c_{n+1}| < \alpha_n/2$ for all large enough n , where $\{b_n\}_{n=1}^\infty$ and $\{\alpha_n\}_{n=1}^\infty$ are sequences of integers as in the assumptions of the theorem. In the middle of the proof, we will have for a large enough N

$$\frac{Bb_N}{\alpha_N} - R_N \text{ is an integer,} \quad (1)$$

where $\{R_n\}_{n=1}^\infty$ is a sequence such that for all $n \geq N$, $|R_n| < 1/4$ and

$$R_{n+1} = \alpha_n R_n - \frac{Bb_{n+1}}{\alpha_{n+1}}. \quad (2)$$

To construct a suitable $\{c_n\}_{n=1}^\infty$, we let c_N be the integer nearest to $(Bb_N)/\alpha_N$ and inductively construct $c_{n+1} = c_n\alpha_n - Bb_n$ (for all $n \geq N$). This was achieved by defining a recursive function in Isabelle/HOL:

```

fun get_c :: "(nat ⇒ int) ⇒ (nat ⇒ int)
           ⇒ int ⇒ nat ⇒ (nat ⇒ int)" where
  "get_c a' b' B N 0 = round (B * b' N / a' N)" |
  "get_c a' b' B N (Suc n) = get_c a' b' B N n * a' (n+N)
                             - B * b' (n+N)"

```

where `round` returns the nearest integer of an input number. By construction, we already have $Bb_n = c_n\alpha_n - c_{n+1}$ so the goal is to show $|c_{n+1}| < \alpha_n/2$ (for all $n \geq N$). To achieve this, we deploy proof by induction to derive

$$c_n - (Bb_n)/\alpha_n = R_n,$$

where (1) has been used for the base case and (2) for the inductive one. We can then close the proof by having

$$\frac{|c_{n+1}|}{\alpha_n} = |R_n| < \frac{1}{2}.$$

The original proof repeatedly invokes variants of (1) and tries to prove c_n is the closest integer to $(Bb_n)/\alpha_n$ for all $n \geq N$ (e.g., (2.7)-(2.9) in the original proof). We believe our altered proof might be more straightforward.

We also noted that in the original proofs of Corollary 3.1 ([13, Corollary 2.10]) and Theorem 3.2 ([13, Theorem 3.1]) certain inequalities required some minor corrections (for details we refer to our formalization [30]), which fortunately did not affect the correctness of the statements nor the original proof structures.

Our formalized versions of the statements of Theorem 3.1 ([13, Theorem 2.1]), Corollary 3.1 ([13, Corollary 2.10]) and Theorem 3.2 ([13, Theorem 3.1]) read as follows:

```

theorem theorem_2_1_Erdos_Straus:
  fixes a b :: "nat $\Rightarrow$ int"
  assumes " $\forall n. a\ n > 0$ " and " $\forall_F n$  in sequentially.  $a\ n > 1$ "
    and " $(\lambda n. |b\ n| / (a\ (n-1)*a\ n)) \longrightarrow 0$ "
  shows " $(\sum n. (b\ n / (\prod i \leq n. a\ i))) \in \mathbb{Q} \iff$ 
    ( $\exists (B::int)>0. \exists c::nat \Rightarrow int.$ 
      ( $\forall_F n$  in sequentially.  $B*b\ n = c\ n * a\ n - c(n+1)$ 
         $\wedge |c(n+1)| < a\ n/2$  ) )"
```

```

corollary corollary_2_10_Erdos_Straus:
  fixes a b :: "nat $\Rightarrow$ int"
  assumes " $\forall n. a\ n > 0$ " and " $\forall_F n$  in sequentially.  $a\ n > 1$ "
    and " $(\lambda n. |b\ n| / (a\ (n-1)*a\ n)) \longrightarrow 0$ "
    and " $\forall_F n$  in sequentially.  $b\ n > 0 \wedge a\ (n+1) \geq a\ n$ "
    and " $\lim (\lambda n. (b(n+1) - b\ n) / a\ n) \leq 0$ "
    and " $\text{convergent } (\lambda n. (b(n+1) - b\ n) / a\ n)$ "
    and " $\liminf (\lambda n. a\ n / b\ n) = 0$ "
  shows " $(\sum n. (b\ n / (\prod i \leq n. a\ i))) \notin \mathbb{Q}$ "
```

```

theorem theorem_3_10_Erdos_Straus:
  fixes a :: "nat  $\Rightarrow$  int"
  assumes " $\forall n. a\ n > 0$ " and "mono a"
    and " $(\lambda n. \text{nth\_prime } n / (a\ n)^2) \longrightarrow 0$ "
    and " $\liminf (\lambda n. a\ n / \text{nth\_prime } n) = 0$ "
  shows " $(\sum n. (\text{nth\_prime } n / (\prod i \leq n. a\ i))) \notin \mathbb{Q}$ "
```

Above, *sequentially* is a *filter* [25] for expressing limits indexed by positive integers. So $\forall_F n$ in *sequentially*. $P\ n$ expresses that $P\ n$ holds for all sufficiently large n .

3.2.1 Calculations with real asymptotics

Asymptotic arguments (e.g. limits and statements of the form “for all sufficiently large n ”) present real difficulties for mechanised proofs. Justifications omitted as obvious in paper proofs need to be written out. For example, in the proof of Theorem 3.2 ([13, Theorem 3.1]) we need to formally prove

$$8B \ln n + 1 < \frac{\sqrt{n}}{4} \quad \text{for all large } n, \quad (3)$$

where B is a positive constant. Since the proof of Eq. (3) involves routine calculations of limits, this step is completely omitted in the paper proof. In Isabelle/HOL, Eq. (3) is encoded as

```
" $\forall_F n$  in sequentially.  $8*B*\ln n + 1 < \sqrt{n}/4$ "
```

Our first attempt resulted in the following proof:

```
have " $\forall_F n$  in sequentially.  $8*B*\ln n + 1 < \sqrt{n}/4$ "
proof -
  have " $(\lambda n. \ln n / \sqrt{n}) \longrightarrow 0$ "
    using lim_ln_over_power[of "1/2"] powr_half_sqrt by simp
  from tendsto_mult_right_zero[OF this]
  have " $(\lambda n. 8 * B * \ln n / \sqrt{n}) \longrightarrow 0$ " by auto
  moreover have " $(\lambda n. 1 / \sqrt{n}) \longrightarrow 0$ "
    apply (rule real_tendsto_divide_at_top)
    by (auto simp add: filterlim_of_nat_at_top sqrt_at_top)
  ultimately have " $(\lambda n. 8*B*\ln n/\sqrt{n} + 1/\sqrt{n}) \longrightarrow 0$ "
    by (auto intro:tendsto_add_zero)
  from tendstoD[OF this, of "1/4"]
  have " $\forall_F n$  in sequentially.  $8*B*\ln n/\sqrt{n} + 1/\sqrt{n} < 1/4$ "
    unfolding abs_less_iff dist_real_def
    by (auto elim!:eventually_mono)
  then show ?thesis
    using eventually_gt_at_top[of 0]
    by (eventually_elim, auto simp:field_simps)
qed
```

The proof is tedious. Let’s work through the text above. We start with primitive properties of the \ln and square root functions,

$$\lim_{n \rightarrow \infty} \frac{\ln n}{\sqrt{n}} = 0 \quad \text{and} \quad \lim_{n \rightarrow \infty} \frac{1}{\sqrt{n}} = 0,$$

written " $(\lambda n. \ln n / \text{sqrt } n) \longrightarrow 0$ " and " $(\lambda n. 1 / \text{sqrt } n) \longrightarrow 0$ " in the script. Then we need to manually transform a property of limits,

$$\lim_{n \rightarrow \infty} \frac{8B \ln n}{\sqrt{n}} + \frac{1}{\sqrt{n}} = 0$$

to the proposition that

$$\frac{8B \ln n}{\sqrt{n}} + \frac{1}{\sqrt{n}} < \frac{1}{4} \quad \text{for all large enough } n,$$

by applying the lemma `tendstoD`:

$$\text{if } \lim_{x \rightarrow F} f(x) = l \text{ and } 0 < e \text{ then eventually}^6 |f(x) - l| < e.$$

Fortunately, Manuel Eberl [8] had just implemented the tactic `real_asymp` that solves Eq. (3) automatically, so we can reduce our initial proof to just one line:

`have "∀F n in sequentially. 8*B*ln n + 1 < sqrt n/4"`
`by real_asymp`

The basic idea behind `real_asymp` is to approximate a continuous function $g: \mathbb{R} \rightarrow \mathbb{R}$ when its argument approaches a limit L using Poincaré expansion (or asymptotic series):

$$\sum_{n=1}^N a_n \varphi_n(x),$$

where $a_n \in \mathbb{R}$ and $\{\varphi_n(x)\}$ is a sequence of continuous functions such that $\varphi_{n+1}(x) \in o(\varphi_n(x))$ for all n as $x \rightarrow L$.⁷ Through arithmetic on the formal series, the asymptotic behaviour of a function at L can then be decided. In general, `real_asymp` handles functions built from basic arithmetic and elementary functions including `exp`, `ln`, and `sin`.

In situations involving limits, due to the presence of bound variables, Sledgehammer's automation helps little, so `real_asymp` makes a real contribution. Such domain-specific automation inspired from computer algebra

⁶as x approaches the filter F

⁷Here, $o(\cdot)$ is the 'small o' notation, and $\varphi_{n+1}(x) \in o(\varphi_n(x))$ as $x \rightarrow L$ is equivalent to $\lim_{x \rightarrow L} \varphi_{n+1}(x)/\varphi_n(x) = 0$.

systems will surely prove to be of great use when mechanising mathematical proofs.

But in general we have found Sledgehammer to be of tremendous value. A telltale sign that it has been used is any occurrence of the proof methods *metis* or *smt*, which are littered throughout our formal proofs [28, 29, 30]. These methods are difficult to use manually, and any instances of them are almost certain to have been generated by Sledgehammer.

3.2.2 Reasoning with prime numbers

Elementary properties concerning prime numbers are often assumed in informal proofs, with supporting calculations largely omitted. Thanks to the mechanised proof of the prime number theorem (PNT) [10] as well as another development on primes [9] from the AFP, basic reasoning with prime numbers is feasible. However, there are still some tedious, repetitious calculations, which could be optimised in the future. For example, the first step in proving Theorem 3.2 ([13, Theorem 3.1]) is to demonstrate that the hypotheses of Theorem 3.1 ([13, Theorem 2.1]) are satisfied by the former's assumptions. Although this step has been completely omitted in the informal proof, we need to explicitly derive $\lim_{n \rightarrow \infty} p_n / (\alpha_{n-1} \alpha_n)$ from $\lim_{n \rightarrow \infty} p_n / \alpha_n^2$, where p_n is the n^{th} prime number and $\{\alpha_n\}_{n=1}^{\infty}$ is a monotonic sequence of positive integers. The derivation works as follows:

$$\lim_{n \rightarrow \infty} \frac{p_n}{\alpha_{n-1} \alpha_n} \leq \lim_{n \rightarrow \infty} \frac{p_n}{\alpha_{n-1}^2} \leq \lim_{n \rightarrow \infty} \frac{2p_{n-1}}{\alpha_{n-1}^2} = 2 \lim_{n \rightarrow \infty} \frac{p_n}{\alpha_n^2} = 0.$$

The second inequality requires a basic property of the prime numbers:

$$\lim_{n \rightarrow \infty} \frac{p_{n+1}}{p_n} = 1. \quad (4)$$

Starting with a form of the PNT, namely $p_n \sim n \ln(n)$, this property alone takes 28 lines of tedious limit calculations similar to those in our first attempt to prove (3). Another example from the proof of Theorem 3.2 ([13, Theorem 3.1]) is the following: considering the interval $[N, 2N)$ for large enough N and given that $c_{n+1} > c_n$ implies $p_{n+1} > p_n + \sqrt{p_n}$, we need to deduce that $c_{n+1} > c_n$ happens for fewer than half of the integers within $[N, 2N)$. The derivation works because

$$\frac{p_{2N} - p_N}{\sqrt{p_N}} < N^{\frac{1}{2} + \epsilon} \quad \text{for all } \epsilon > 0 \text{ and all large enough } N, \quad (5)$$

and $N^{\frac{1}{2}+\epsilon} < N/2$ when $\epsilon < 1/2$ and N is large enough. Property (5) is formulated in Isabelle/HOL as

```
lemma nth_prime_double_sqrt_less:
  assumes "ε > 0"
  shows "∀F n in sequentially. (nth_prime (2*n) - nth_prime n)
    / sqrt (nth_prime n) < n powr (1/2 + ε)"
```

where we remind that $\forall_F n$ in *sequentially* stands for the “large enough” quantifier. Despite (5) being merely claimed without proof in the paper, it took us 56 lines to prove it in Isabelle/HOL starting from the PNT. In short, we are happy that properties related to the prime numbers are now viable in Isabelle/HOL, but we wish we could derive their asymptotic properties like (4) and (5) with fewer tears. Automating this process through the tactic *real_asymp* is desirable but the tactic does not yet support the function *nth_prime*: in contrast to elementary functions like *exp*, *nth_prime* cannot be asymptotically expanded to an arbitrary order. Incorporating *nth_prime* into *real_asymp* may require special treatment.

4 Irrational rapidly convergent series

4.1 Results formalized

The following results are by Hančl [23].

Theorem 4.1 (Hančl [23, Theorem 3]) *Let $A > 1$ be a real number. Let $\{d_n\}_{n=1}^\infty$ be a sequence of real numbers greater than one. Let $\{\alpha_n\}_{n=1}^\infty$ and $\{b_n\}_{n=1}^\infty$ be sequences of positive integers such that $\lim_{n \rightarrow \infty} \alpha_n^{\frac{1}{2^n}} = A$ and for all sufficiently large n*

$$\frac{A}{\alpha_n^{\frac{1}{2^n}}} > \prod_{j=n}^{\infty} d_j \quad \text{and} \quad \lim_{n \rightarrow \infty} \frac{d_n^{2^n}}{b_n} = \infty.$$

Then the sum

$$\sum_{n=1}^{\infty} \frac{b_n}{\alpha_n}$$

is irrational.

Corollary 4.1 (Hančl [23, Corollary 2]) *Let $A > 1$ and let $\{\alpha_n\}_{n=1}^\infty$ and $\{b_n\}_{n=1}^\infty$ be sequences of positive integers such that $\lim_{n \rightarrow \infty} \alpha_n^{\frac{1}{2^n}} = A$. Then, assuming that for every sufficiently large positive integer n , $\alpha_n^{\frac{1}{2^n}} (1 + 4(2/3)^n) \leq A$ and $b_n \leq 2^{(4/3)^{n-1}}$, the sum $\sum_{n=1}^\infty b_n/\alpha_n$ is irrational.*

This corollary follows directly from the theorem by an appropriate choice of the sequence $\{d_n\}_{n=1}^\infty$, in particular by choosing $d_n = 1 + (2/3)^n$ so that the theorem can be directly applied.

4.2 On the formalization

Our full formalization is online on the AFP [28]. This was the first work in this area that we formalized, starting in early 2018. We chose this research paper as the proofs are interesting but not too complicated, and even though this is a topic in number theory, they are based on fundamental material from analysis. Of course, we also had to show quite a few auxiliary facts, including lemmas on summability criteria for series. For instance, we modified the classic ratio test for series (which was already available in the HOL Library):

```
lemma summable_ratio_test:
  fixes c :: real and N :: nat
    and f :: "nat ⇒ 'a :: real_normed_vector"
  assumes "c < 1" and "∀n ≥ N. norm (f (Suc n)) ≤ c * norm (f n)"
  shows "summable f"
```

The version modified in terms of limits is given below:

```
lemma summable_ratio_test_tendsto:
  fixes c :: real and f :: "nat ⇒ 'a :: real_normed_vector"
  assumes "c < 1" and "∀n. f n ≠ 0"
    and tendsto_c: "(λn. norm (f (Suc n)) / norm (f n)) ⟶ c"
  shows "summable f"
proof -
  from ⟨c<1⟩ tendsto_c
  obtain N where N_dist:
    "∀n ≥ N. dist (norm (f (Suc n)) / norm (f n)) c < (1-c)/2"
  unfolding tendsto_iff eventually_sequentially
  by (meson diff_gt_0_iff_gt zero_less_divide_iff zero_less_numeral)
  have "norm (f (Suc n)) / norm (f n) ≤ (1+c)/2" when "n ≥ N" for n
  using N_dist[rule_format,OF ⟨n ≥ N⟩] ⟨c<1⟩
```

```

    by (auto simp add:field_simps dist_norm,argo)
  then have "norm (f (Suc n)) ≤ (1+c)/2 * norm (f n)"
    when "n≥N" for n
    using ⟨n≥N⟩ ⟨∀n. f n ≠ 0⟩ by (auto simp add:divide_simps)
  moreover have "(1+c)/2 < 1" using ⟨c<1⟩ by auto
  ultimately show "summable f"
    using summable_ratio_test by blast
qed

```

Our formalized version of Theorem 4.1 ([23, Theorem 3]) is presented below. We include its formalized proof in the Appendix. The reader will notice that the formal proof is only around 50 lines of Isar code; however, it comes after about 800 lines of auxiliary lemmas that are required for this specific proof. As in the original—and after having established the summability of the series in question—the proof is by contradiction. Assuming that the sum is rational, for a quantity $\text{ALPHA}(n)$ we show that $\text{ALPHA}(n) \geq 1$ for all $n \in \mathbb{N}$. After that, we show the existence of an $n \in \mathbb{N}$ for which $\text{ALPHA}(n) < 1$, a contradiction: hence, the sum of the series is irrational. As already mentioned, Isar admits structured proofs, which are more readable than the usual sequence of proof commands. In our proof, the structure of a proof by contradiction is visible:

```

  proof(rule ccontr) [...] show False [...] qed.

```

theorem Hanc13:

```

  fixes d :: "nat ⇒ real" and a b :: "nat ⇒ int"
  assumes "A > 1" and d: "∀n. d n > 1"
    and a: "∀n. a n > 0" and b: "∀n. b n > 0" and "s>0"
    and "(λn. (a n) powr(1 / of_int(2^n))) → A"
    and "∀n ≥ s. A / (a n) powr (1 / of_int(2^n)) > (∏ j. d (n+j))"
    and "LIM n sequentially. d n ^ 2 ^ n / b n :=> at_top"
    and "convergent_prod d"
  shows "(∑ n. b n / a n) ∉ ℚ

```

Our formalized version of Corollary 4.1 ([23, Corollary 2]) is as follows:

corollary Hanc13corollary:

```

  fixes A::real and a b :: "nat ⇒ int"
  assumes "A > 1" and a: "∀n. a n > 0" and b: "∀n. b n > 0"
    and "(λn. (a n) powr(1 / of_int(2^n))) → A"
    and "∀n ≥ 6. a n powr(1 / of_int (2^n)) * (1 + 4*(2/3)^n) ≤ A

```

$\wedge b_n \leq 2 \text{ powr } (4/3)^{(n-1)}$

shows " $(\sum_n. b_n / a_n) \notin \mathbb{Q}$ "

It is interesting to note that while in Corollary 4.1 ([23, Corollary 2]) we have an assumption that holds “for all sufficiently large positive integer n ”, in our formalized version above this assumption is actually specified as “for all $n \geq 6$ ”. In our formalized version of Theorem 4.1 ([23, Theorem 3]) above, “for all sufficiently large n ” had been written simply as “ $\forall n \geq s$ ” where $s > 0$ is some fixed, unknown number. Statements of the form “for all sufficiently large” can also be encoded in a more abstract fashion using the keyword **eventually**, which does not require an additional variable.

4.2.1 Infinite Products

This material requires the notion of an infinite product, which was then (early 2018) not available in Isabelle except in a small development by Manuel Eberl. Infinite sums were available, and because we were only dealing with positive reals, the infinite products could be reduced to infinite sums via logarithms. However, since one objective of this work was to identify and fill gaps in the libraries, we decided to formalize infinite products directly. We extended Eberl’s development to a comprehensive *Infinite_Products* library and included it in the next Isabelle release (Isabelle 2018).

Consider the following technical lemma, where we make use of \prod :

lemma show8:

```

fixes  $d :: \text{"nat} \Rightarrow \text{real"}$  and  $a :: \text{"nat} \Rightarrow \text{int"}$  and  $s k :: \text{nat}$ 
assumes " $A > 1$ " and  $d: \forall n. d\ n > 1$ " and " $\forall n. a\ n > 0$ " and " $s > 0$ "
and "convergent_prod  $d$ "
and " $\forall n \geq s. A / (a\ n) \text{ powr } (1 / (2 :: \text{int})^n) > (\prod j. d\ (n + j))$ "
shows " $\forall n \geq s. (\prod j. d\ (j + n)) < A$ 
        / ( $\text{MAX } j \in \{s..n\}. a\ j \text{ powr } (1 / (2 :: \text{int})^j)$ )"
```

This had been originally formulated using \sum via the natural logarithm in the first version, so the conclusion instead read

```

shows " $\forall n \geq s. \exp (\sum j. \ln(d\ (j + n))) < A$ 
        / ( $\text{MAX } j \in \{s..n\}. a\ j \text{ powr } (1 / (2 :: \text{int})^j)$ )"
```

By convention, infinite products are defined in two stages [1, p. 241].

- (1) For nonzero complex numbers, the infinite product $\prod_{k=0}^{\infty} u_k$ is defined to *converge* if the finite partial products converge to a nonzero limit. If that limit is zero, then the infinite product *diverges* to zero.
- (2) If finitely many of the u_k equal zero and the infinite product of the nonzero terms converges in the sense of (1), then $\prod_{k=0}^{\infty} u_k$ converges to zero.

One advantage of this approach is that $\prod_{k=0}^{\infty} u_k = u_0 \prod_{k=1}^{\infty} u_k$ and similar identities hold.

5 The transcendence of certain infinite series

5.1 Results formalized

The following results are by Hančl and Rucki in [24].

Theorem 5.1 (Hančl and Rucki [24, Theorem 2.1]) *Let δ be a positive real number. Let $\{\alpha_n\}_{n=1}^{\infty}$ and $\{b_n\}_{n=1}^{\infty}$ be sequences of positive integers such that*

$$\limsup_{n \rightarrow \infty} \frac{\alpha_{n+1}}{(\prod_{i=1}^n \alpha_i)^{2+\delta}} \cdot \frac{1}{b_{n+1}} = \infty$$

and

$$\liminf_{n \rightarrow \infty} \frac{\alpha_{n+1}}{\alpha_n} \cdot \frac{b_n}{b_{n+1}} > 1.$$

Then the sum

$$\sum_{n=1}^{\infty} \frac{b_n}{\alpha_n}$$

is transcendental.

Theorem 5.2 (Hančl and Rucki [24, Theorem 2.2]) *Let δ and ϵ be positive real numbers. Let $\{\alpha_n\}_{n=1}^{\infty}$ and $\{b_n\}_{n=1}^{\infty}$ be sequences of positive integers such that*

$$\limsup_{n \rightarrow \infty} \frac{\alpha_{n+1}}{(\prod_{i=1}^n \alpha_i)^{2+2/\epsilon+\delta}} \cdot \frac{1}{b_{n+1}} = \infty$$

and for every sufficiently large n

$$\sqrt[1+\epsilon]{\frac{\alpha_{n+1}}{b_{n+1}}} \geq \sqrt[1+\epsilon]{\frac{\alpha_n}{b_n}} + 1.$$

Then the sum

$$\sum_{n=1}^{\infty} \frac{b_n}{\alpha_n}$$

is transcendental.

The two theorems above are in fact corollaries of Roth's celebrated result on rational approximations to algebraic numbers from 1955:

Theorem 5.3 (Roth [38]) *Let a be any algebraic number, not rational. If the inequality*

$$\left| a - \frac{p}{q} \right| < \frac{1}{q^\kappa}$$

has infinitely many solutions in coprime integers p and q where $q > 0$, then $\kappa \leq 2$.

In particular, the two theorems above by Hančl and Rucki [24] asserting the transcendence of the sum of certain series are shown by finding infinitely many such integer solutions of the inequality with a the sum of the series in question and for some $\kappa > 2$.

5.2 On the formalization

Our full formalization is online [29].⁸ In this work, too, we had to fill in certain arguments that had been omitted in the original paper, for example to first show that the series in question were summable.

5.2.1 Implementing Roth's Theorem

As already mentioned, for the proofs of Theorems 5.1 ([24, Theorem 2.1]) and 5.2 ([24, Theorem 2.2]), we applied Theorem 5.3: Roth's theorem on rational approximations to algebraic numbers [38]. The proof of this celebrated theorem is long and elaborate; it has not been formalized in any proof assistant, to our knowledge. In our formalization we have thus used the statement of Roth's theorem merely as an assumption, instead of formalizing Roth's proof itself beforehand. Roth's theorem was implemented within a *locale* as follows:

⁸The reader may notice that this development too seems to depend on the AFP entry for the Prime Number Theorem [10]. However we only use some technical lemmas from that development. Primes play no role here.

locale *RothsTheorem* =

assumes *RothsTheorem*:

" $\forall \xi \kappa. \text{algebraic } \xi \wedge \xi \notin \mathbb{Q} \wedge \text{infinite } \{(p,q). (q::\text{int}) > 0 \wedge \text{coprime } p \ q \wedge |\xi - p/q| < 1/q \text{ powr } \kappa\} \rightarrow \kappa \leq 2$ "

A locale collects parameters and assumptions, which it packages as a context in which to work. This locale simply packages the assumption that Roth's theorem is true. It is a safer option than assuming the theorem as an axiom. Theorems 5.1 ([24, Theorem 2.1]) and 5.2 ([24, Theorem 2.2]) were then formalized within this locale:

theorem (in *RothsTheorem*) *HanclRucki1*:

fixes $a \ b :: \text{"nat} \Rightarrow \text{int"}$ **and** $\delta :: \text{real}$

assumes " $\forall k. a \ k > 0$ " **and** " $\forall k. b \ k > 0$ " **and** " $\delta > 0$ "

and " $\text{limsup } (\lambda k. a(k+1) /$

$(\prod_{i=0..k} a \ i) \text{ powr } (2+\delta) * (1 / b(k+1))) = \infty$ "

and " $\text{liminf } (\lambda k. a \ (k+1) / a \ k * b \ k / b \ (k+1)) > 1$ "

shows " $\neg \text{algebraic } (\sum k. b \ k / a \ k)$ "

theorem (in *RothsTheorem*) *HanclRucki2*:

fixes $a \ b :: \text{"nat} \Rightarrow \text{int"}$ **and** $\delta \ \varepsilon :: \text{real}$ **and** $t :: \text{nat}$

assumes " $\forall k. a \ k > 0$ " **and** " $\forall k. b \ k > 0$ " **and** " $\delta > 0$ " " $\varepsilon > 0$ "

and " $\text{limsup } (\lambda k. a(k+1) / (\prod_{i=0..k} a \ i) \text{ powr } (2 + 2/\varepsilon + \delta)$

$* (1 / b(k+1))) = \infty$ "

and " $\forall k \geq t. (a \ (k+1) / b \ (k+1)) \text{ powr } (1 / (1+\varepsilon))$

$\geq (a \ k / b \ k) \text{ powr } (1 / (1+\varepsilon)) + 1$ "

shows " $\neg \text{algebraic } (\sum k. b \ k / a \ k)$ "

Assuming a key theorem whose proof has not been formalized could be seen as compromising the vision of absolute correctness, which demands a bottom-up approach: to formalize the proofs of all prerequisites. However, ours is a more realistic approach that reflects actual mathematical practice. It still guarantees the correctness of the arguments we formalized, which assume Roth's theorem. Normally, whenever mathematicians make use of a theorem in their work, they do not prove it themselves all over again but they trust the proof in the literature. But when writing a proof with a proof assistant, one would ideally expect that every result used would also be *formally* verified as well. Here, however, Roth's theorem is used even though its proof hasn't been formalized; we just trust its *informal* proof [38], just like a traditional mathematician would do.

Our approach has the significant advantage of reaching the formalization of more advanced mathematics faster. Nevertheless, care must be taken to avoid propagating errors in the literature. We recommend that a formally unverified theorem should be taken as an assumption within a formalization only if it is a fundamental result that has been checked by many mathematicians in the past. Roth’s famous theorem can be considered safe in this sense. And all such assumptions must be declared openly [27, Section 5].

5.2.2 A small issue in the original proof

The process of the formalization helped to reveal a slight mistake in one of the original proofs. It puzzled us for some time, although the eventual fix was straightforward. In the proof of Theorem 5.1 [24, page 534] it is claimed that from the assumptions it follows that for each real number $A > 1$, there exists a positive integer k_0 such that for all $k > k_0$,

$$\frac{1}{A} \cdot \frac{b_k}{\alpha_k} > \frac{b_{k+1}}{\alpha_{k+1}}.$$

During the formalization process, we noticed that there is a problem with this claim. Consider the counterexample where we set $\alpha_{k+1} = k(\prod_{i=1}^k \alpha_i)^{\lceil 2+\delta \rceil}$ if k is odd, and $\alpha_{k+1} = 2\alpha_k$ otherwise, with $b_k = 1$ for all k . To resolve this, we had suggested a slightly modified version of the original statement and a different proof. However, this turned out to be unnecessary; the authors suggested to us via email that the problem could be resolved by replacing “for each real number $A > 1$ ” with “there exists a real number $A > 1$ ” above. This suggestion repaired the proof, and the rest of our formalization [29] followed the original proof [24] without any further problems.

6 On the de Bruijn factors

The de Bruijn factor, introduced in 1977 by L. S. van Benthem Jutting [26], is the ratio of the size of a formalization (in symbols) to the size of the original mathematical text (in words). Freek Wiedijk [40] recommends making this more precise by comparing the number of bytes in the computer encodings (using the \LaTeX source of the mathematics), even compressing both files to ensure independence from such factors as the lengths of identifiers. Such precision is of questionable value, given the enormous variations in the density of mathematical texts.

Here we only give approximate estimates of the de Bruijn factors for our formalizations, counting the number of lines in our formalizations and in the respective published informal proofs. We consider the entire amount of the material (i.e. statements together with their proofs as well as corollaries together with their proofs) in each formalization work. Given our formalization [30] which spans around 1960 lines, we estimate the de Bruijn factor for Theorem 3.1 ([13, Theorem 2.1]), Corollary 3.1 ([13, Corollary 2.10]) and Theorem 3.2 ([13, Theorem 3.1]) and their respective proofs altogether to be around 25. Given our formalization [28] which spans around 1054 lines, we estimate the de Bruijn factor for Theorem 4.1 ([23, Theorem 3]), Corollary 4.1 ([23, Corollary 2]) and their respective proofs altogether to be around 21. Finally, given our formalization [29] which spans around 990 lines, we estimate the de Bruijn factor for Theorem 5.1 ([24, Theorem 2.1]) and Theorem 5.2 ([24, Theorem 2.2]) and their respective proofs altogether to be around 13. These de Bruijn factors can be considered high—Wiedijk reports lower factors for some formalizations in HOL [41]—because we had to fill in a considerable amount of intermediate arguments which had not been made explicit in the original proofs, especially in the case of Erdős–Straus [13].

7 Conclusion

We have formalized, in Isabelle/HOL [28, 29, 30], several results on irrationality and transcendence criteria for infinite series from three research papers: by Erdős–Straus (1974) [13], Hančl (2002) [23] and Hančl–Rucki (2005) [24].

We formalized results from mainstream journal papers in their original, unpolished form. Landmarks in the world of formalized mathematics include proving the four-colour theorem in Coq by Georges Gonthier [15], the Kepler conjecture [18] in HOL Light and Isabelle by Thomas C. Hales et al. [19] and, more recently, the solution to the Cap Set Problem in Lean by Dahmen, Hölzl and Lewis [7], the formalization of perfectoid spaces in Lean by Buzzard, Commelin and Massot [6] and the formalization of Grothendieck’s Schemes in Isabelle/HOL by Bordg, Paulson and Li [3, 4]. More in the spirit of our work is the formalization in Isabelle/HOL by Gouëzel of a recent paper by Shchur in the theory of Gromov-hyperbolic spaces, where a mistake in the original paper was moreover discovered through the formalization process [16].

Summarizing the lessons that we learned through our formalization, we

can now revisit the questions posed in the Introduction:

- The amount of preliminary content in the current Isabelle/HOL libraries was indeed satisfactory. We had a priori chosen material that we expected would be within reach, and indeed, we didn't run into significant deficiencies in the library. Some prerequisites that had to be added were a detailed formalization of infinite products, a modification of the ratio test for series and some facts about prime numbers. All these additions were doable within a reasonable amount of time and we can expect that they will be used in many other contexts. Our formalizations were thus an incentive to enrich the library.
- In general we have found Sledgehammer's automation to be of immense help, as hinted by the many occurrences throughout our formal proofs of proof methods such as *metis* and *smt*, which are generated by Sledgehammer.
- As discussed in Section 6, the formal proofs were significantly longer than the original, which is reflected by the fairly high de Bruijn factors (approximately 25, 21 and 13 for the material [13, 23, 24] respectively). This was certainly not unexpected.
- We did not discover any major issues in the original proofs. We did however need to fill in some missing steps or calculations, some of which were proofs of summability of infinite series. These omissions contributed to the high de Bruijn factors. Regarding Erdős–Straus [13], one of the proofs was slightly reformulated, while in other proofs a few inequalities required some minor corrections that did not, however, affect the correctness of the statements or the general proof structure. In Hančl–Rucki [24], a slight mistake was easily fixed thanks to a hint from the authors. Our detailed, complete formal proofs may even be useful to the readers who wish to study the original proofs.

Overall, we found the formalization process to be interesting and educational. We hope that our feedback may be useful both to experienced proof assistant users and to mathematicians interested in attempting to formalize some research work for the first time.

Acknowledgements. The authors thank the ERC for their support through the ERC Advanced Grant ALEXANDRIA (Project GA 742178). We thank

the anonymous reviewers for their useful suggestions that improved our exposition. We also thank Jaroslav Hančl and Pavel Rucki for a helpful email suggesting a fix for the slight mistake in their proof [24, Theorem 2.1], Iosif Pinelis for his useful explanation on MathOverflow regarding an argument in the proof of Theorem 2.2 in the same paper⁹ and Anthony Bordg for a discussion on the necessity of formalizing advanced mathematics even if the proofs of the prerequisites are not formalized.

Appendix

Our formalized version of Theorem 4.1 ([23, Theorem 3]) along with its formalized proof as in [28] is presented below:

theorem Hanc13:

```

fixes d :: "nat ⇒ real " and a b :: "nat ⇒ int" and A::real and s::nat
assumes "A > 1" and d:"∀n. d n > 1" and a:"∀n. a n > 0"
      and b:"∀n. b n > 0" and "s > 0"
      and assu1: "( (λ n. a n powr (1/(2::int)^n)) → A) sequentially"
      and assu2: "∀n ≥ s. A / a n powr (1/(2::int)^n) > (∏ j. d (n + j))"
      and assu3: "LIM n sequentially. d n ^ 2 ^ n / b n :> at_top"
      and "convergent_prod d"
shows "(∑ n. b n / a n) ∉ ℚ"
proof(rule ccontr)
  assume asm:"¬ ((∑ n. b n / a n) ∉ ℚ)"
  have ab_sum:"summable (λj. b j / a j)"
    using issummable[OF ⟨A>1⟩ d a b assu1 assu3 ⟨convergent_prod d⟩] .
  obtain p q ::int where "q>0"
    and pq_def:"(λ n. b (n+1) / a (n+1)) sums (p/q)"
  proof -
    from asm have "(∑ n. b n / a n) ∈ ℚ" by auto
    then have "(∑ n. b (n+1) / a (n+1)) ∈ ℚ"
      by (subst suminf_minus_initial_segment[OF ab_sum,of 1]) auto
    then obtain p' q' ::int where "q' ≠ 0"
      and pq_def:"(λ n. b (n+1) / a (n+1) ) sums (p'/q')"
      unfolding Rats_eq_int_div_int
      using summable_ignore_initial_segment[OF ab_sum,of 1,THEN summable_sums]
      by force
    define p q where "p=(if q'<0 then - p' else p')"
```

⁹<https://mathoverflow.net/questions/323069/why-is-this-series-summable>

```

        and "q=(if q'<0 then - q' else q')"
    have "p'/q'=p/q" and "q>0"
        using ⟨q'≠0⟩ unfolding p_def q_def by auto
    then show ?thesis using that[of q p] pq_def by auto
qed

define ALPHA where
    "ALPHA = (λn. q * (∏j=1..n. a j) * (∑j. b (j+n+1) / a (j+n+1)))"
have "ALPHA n ≥ 1" for n
proof -
    have "(∑ n. b (n+1) / a (n+1)) > 0"
        apply (rule suminf_pos)
        using summable_ignore_initial_segment[OF ab_sum,of 1] a b by auto
    then have "p/q > 0" using sums_unique[OF pq_def,symmetric] by auto
    then have "q≥1" and "p≥1" using ⟨q>0⟩ by (auto simp add:divide_simps)
    moreover have "∀n. 1 ≤ a n" and "∀n. 1 ≤ b n"
        using a b by (auto simp add: int_one_le_iff_zero_less)
    ultimately show ?thesis unfolding ALPHA_def
        using show7[OF _ _ _ pq_def] by auto
qed
moreover have "∃n. ALPHA n < 1" unfolding ALPHA_def
proof (rule lasttoshow[OF d a ⟨s>0⟩ ⟨q>0⟩ ⟨A>1⟩ b _ assu3])
    show "∀F n in sequentially. (∑j. b (n + j) / a(n + j))
        ≤ 2 * b n / a n"
        using show5[OF ⟨A>1⟩ d a b assu1 assu3 ⟨convergent_prod d⟩] by simp
    show "∀m≥s. ∃n≥m. d (n + 1)
        * (MAX j∈{s..n}. a j powr (1 / (2::int) ^ j))
        ≤ a (n + 1) powr (1 / (2::int) ^ (n + 1))"
        apply (rule show9[OF ⟨A>1⟩ d a ⟨s>0⟩ assu1 ⟨convergent_prod d⟩])
        using show8[OF ⟨A>1⟩ d a ⟨s>0⟩ ⟨convergent_prod d⟩ assu2]
        by (simp add:algebra_simps)
qed
ultimately show False using not_le by blast
qed

```

References

- [1] Bak, J. and Newman, D. J., *Complex Analysis*, Springer, 3rd edition (2010).

- [2] Blanchette, J. C., Kaliszyk, C., Paulson, L. C. and Urban, C., *Hammering towards QED*, Journal of Formalized Reasoning, 9(1):101–148 (2016).
- [3] Bordg, A., Paulson C., L. and Li, W., *Grothendieck’s Schemes in Algebraic Geometry*, Archive of Formal Proofs (2021), formal proof development, available on https://isa-afp.org/entries/Grothendieck_Schemes.html.
- [4] Bordg, A., Paulson C., L. and Li, W., *Simple Type Theory is not too Simple: Grothendieck’s Schemes without Dependent Types*, arXiv:2104.09366 (2021).
- [5] Boyer, R. et al., *The QED Manifesto*, in: A. Bundy (ed.), Automated Deduction – CADE 12, LNAI 814, Springer-Verlag, 238–251 (1994).
- [6] Buzzard, K., Commelin, J. and Massot, P., *Formalising perfectoid spaces*, arXiv:1910.12320, in CPP 2020: Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs, Pages 299–312.
- [7] Dahmen, S. R., Hölzl, J. and Lewis, R. Y., *Formalizing the Solution to the Cap Set Problem*, arXiv:1907.01449v1, Proceedings of Interactive Theorem Proving (ITP) 2019.
- [8] Eberl, M. *Verified real asymptotics in Isabelle/HOL*. In Proceedings of International Symposium on Symbolic and Algebraic Computation (2019).
- [9] Eberl, M. *Elementary Facts About the Distribution of Primes*, Archive of Formal Proofs (2019), formal proof development, available on https://www.isa-afp.org/entries/Prime_Distribution_Elementary.html.
- [10] Eberl, M. and Paulson, L.C., *The Prime Number Theorem*, Archive of Formal Proofs (2018), formal proof development, available on https://www.isa-afp.org/entries/Prime_Number_Theorem.html.
- [11] Erdős, P., *Sur certaines series à valeur irrationnelle*, Enseignement Math., Vol.4, pp.93-100, (1958).
- [12] Erdős, P., *Some problems and results on the irrationality of the sum of infinite series*, J. Math. Sci., Vol. 10, pp.1-7 (1975).

- [13] Erdős, P. and Straus, E.G., *On the irrationality of certain series*, Pacific Journal of Mathematics, Vol. 55, No 1, pp. 85–92 (1974).
- [14] Erdős, P. and Straus, E.G., *Some number theoretic results*, Pacific Journal of Mathematics, Vol. 36, pp. 635–646 (1971).
- [15] Gonthier, G., *Formal Proof–The Four-Color Theorem*, Notices of the American Mathematical Society, 55 (11): 1382–1393 (2008).
- [16] Gouëzel, S. and Shchur, V., *A corrected quantitative version of the Morse lemma*, Journal of Functional Analysis, Vol. 277, Issue 4, 1258–1268 (2019).
- [17] Gowers, W.T., *Rough Structure and Classification*, in: Alon N., Bourgain J., Connes A., Gromov M., Milman V. (eds.) Visions in Mathematics. Modern Birkhäuser Classics. Birkhäuser Basel (2010).
- [18] Hales, T. C., *A proof of the Kepler conjecture*, Annals of Mathematics. 162 (3): 1065–1185 (2005).
- [19] Hales, T., Adams, M., Bauer, G., Dang, T. D., Harrison, J., Hoang, L. T., Kaliszyk, C., Magron, V., McLaughlin, S., Nguyen, T. T., Nguyen, Q. T., Nipkow, T., Obua, S., Pleso, J., Rute, J., Solovyev, A., Ta, T. H. A., Tran, N. T., Trieu, T. D., Urban, J., Vu, K., Zumkeller, R., *A Formal Proof of the Kepler Conjecture*, Forum of Mathematics, Pi., Vol. 5, e2 (2017).
- [20] Hančl, J., *Criterion for irrational sequences*, J. Number Theory, Vol. 43, No. 1, pp. 88-92 (1993).
- [21] Hančl, J., *Transcendental Sequences*, Math. Slov., Vol. 46, pp.177-179 (1996).
- [22] Hančl, J., *Two criteria for transcendental sequences*, Le Matematiche, Vol. 56, pp. 149-160 (2001).
- [23] Hančl, J., *Irrational rapidly convergent series*, Rend. Sem. Mat. Univ. Padova, Vol. 107, pp. 225–231 (2002).
- [24] Hančl, J. and Rucki, P., *The transcendence of certain infinite series*, Rocky Mountain Journal of Mathematics, 35(2), pp. 531–537 (2005).

- [25] Hölzl, J., Immler, F. and Huffman, B. *Type classes and filters for mathematical analysis in Isabelle/HOL*. In International Conference on Interactive Theorem Proving 2013 Jul 22 (pp. 279-294). Springer, Berlin, Heidelberg.
- [26] Jutting, L.S. van Benthem. *Checking Landau's "Grundlagen" in the AUTOMATH System*. PhD thesis, Eindhoven University of Technology (1977). <https://doi.org/10.6100/IR23183>.
- [27] Koutsoukou-Argyraki, A., *Formalising Mathematics – in Praxis; A Mathematician's First Experiences with Isabelle/HOL and the Why and How of Getting Started*, Jahresber. Dtsch. Math. Ver. 123, pp. 3–26(2021). <https://doi.org/10.1365/s13291-020-00221-1>
- [28] Koutsoukou-Argyraki, A. and Li, W., *Irrational rapidly convergent series*, Archive of Formal Proofs (2018), formal proof development, available on https://www.isa-afp.org/entries/Irrationality_J_Hancl.html.
- [29] Koutsoukou-Argyraki, A. and Li, W., *The transcendence of certain infinite series*, Archive of Formal Proofs (2019), formal proof development, available on https://www.isa-afp.org/entries/Transcendence_Series_Hancl_Rucki.html.
- [30] Koutsoukou-Argyraki, A. and Li, W., *Irrationality Criteria for Series by Erdős and Straus*, Archive of Formal Proofs (2020), formal proof development, available on https://www.isa-afp.org/entries/Irrational_Series_Erdos_Straus.html.
- [31] Mahler, K., *Lectures on transcendental numbers*, Lecture Notes in Mathematics 546, Springer, (1976).
- [32] Nipkow, T., Paulson, L. C., and Wenzel, M., *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*, Lecture Notes in Computer Science, Vol. 2283, Springer-Verlag (2002).
- [33] Nishioka, K., *Mahler functions and transcendence*, Lecture Notes in Mathematics 1631, Springer, (1996).
- [34] Nyblom, M.A., *A theorem on transcendence of infinite series*, Rocky Mountain Journal of Mathematics, Vol. 30 pp. 1111-1120 (2000).

- [35] Nyblom, M.A., *A theorem on transcendence of infinite series II*, J. Number Theory Vol. 91, pp.71-80 (2001).
- [36] Paulson, L. C., *The Foundation of a Generic Theorem Prover*, Journal of Automated Reasoning, Vol. 5, Issue 3: 363–397, Kluwer Academic Publishers (1989).
- [37] Paulson, L. C., *ALEXANDRIA: Large-scale formal proof for the working mathematician*, project description, (12 Nov. 2018), available on <http://www.cl.cam.ac.uk/~lp15/Grants/Alexandria/>.
- [38] Roth, K.F., *Rational approximations to algebraic numbers*, Mathematica Vol. 2, Part 1, No. 3, 1–20 (1955).
- [39] Sándor, J., *Some classes of irrational numbers*, Studia, Babeş-Bolyai, Mathematica, Vol. 29, pp.3-12 (1994).
- [40] Wiedijk, F., *The De Bruijn Factor*, Technical report, Department of Computer Science Nijmegen University, P.O. Box 9010, 6500 GL Nijmegen, Netherlands (2000), Online at <https://www.cs.ru.nl/~freek/factor/factor.pdf>.
- [41] Wiedijk, F., *The De Bruijn Factor* (webpage), accessed on 8 Apr. 2021, available on <http://www.cs.ru.nl/~freek/factor/>.